pragma solidity ^0.8.23;
// SPDX-License-Identifier: MIT

// ChekorX

// contract code starts on line 1300

// apologies for the monolithic single file.
// gotta figure out how to do multi file verification.

// OpenZeppelin Contracts (last updated v4.5.0) (token/ERC721/ERC721.sol)
// OpenZeppelin Contracts v4.4.1 (token/ERC721/IERC721.sol)
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;
}

// OpenZeppelin Contracts v4.4.1 (token/ERC721/IERC721Receiver.sol)
/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/IERC721Metadata.sol)
/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}

// OpenZeppelin Contracts (last updated v4.5.0) (utils/Address.sol)
/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)
/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// OpenZeppelin Contracts v4.4.1 (utils/Strings.sol)
/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}

// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)
/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
contract ERC721 is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;
    using Strings for uint256;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to owner address
    mapping(uint256 => address) private _owners;

    // Mapping owner address to token count
    mapping(address => uint256) private _balances;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view virtual override returns (uint256) {
        require(owner != address(0), "ERC721: balance query for the zero address");
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view virtual override returns (address) {
        address owner = _owners[tokenId];
        require(owner != address(0), "ERC721: owner query for nonexistent token");
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        require(_exists(tokenId), "ERC721Metadata: URI query for nonexistent token");

        string memory baseURI = _baseURI();
        return bytes(baseURI).length > 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : "";
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overriden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return "";
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not owner nor approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view virtual override returns (address) {
        require(_exists(tokenId), "ERC721: approved query for nonexistent token");

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        _setApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) public virtual override {
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");
        _safeTransfer(from, to, tokenId, _data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `_data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _transfer(from, to, tokenId);
        require(_checkOnERC721Received(from, to, tokenId, _data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _owners[tokenId] != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId) internal view virtual returns (bool) {
        require(_exists(tokenId), "ERC721: operator query for nonexistent token");
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner || getApproved(tokenId) == spender || isApprovedForAll(owner, spender));
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, _data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId);

        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);

        _afterTokenTransfer(address(0), to, tokenId);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId);

        // Clear approvals
        _approve(address(0), tokenId);

        _balances[owner] -= 1;
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);

        _afterTokenTransfer(owner, address(0), tokenId);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer from incorrect owner");
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId);

        _balances[from] -= 1;
        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);

        _afterTokenTransfer(from, to, tokenId);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits a {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Approve `operator` to operate on all of `owner` tokens
     *
     * Emits a {ApprovalForAll} event.
     */
    function _setApprovalForAll(
        address owner,
        address operator,
        bool approved
    ) internal virtual {
        require(owner != operator, "ERC721: approve to caller");
        _operatorApprovals[owner][operator] = approved;
        emit ApprovalForAll(owner, operator, approved);
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param _data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) private returns (bool) {
        if (to.isContract()) {
            try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, _data) returns (bytes4 retval) {
                return retval == IERC721Receiver.onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert("ERC721: transfer to non ERC721Receiver implementer");
                } else {
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}
}

// OpenZeppelin Contracts v4.4.1 (utils/Counters.sol)
/**
 * @title Counters
 * @author Matt Condon (@shrugs)
 * @dev Provides counters that can only be incremented, decremented or reset. This can be used e.g. to track the number
 * of elements in a mapping, issuing ERC721 ids, or counting request ids.
 *
 * Include with `using Counters for Counters.Counter;`
 */
library Counters {
    struct Counter {
        // This variable should never be directly accessed by users of the library: interactions must be restricted to
        // the library's function. As of Solidity v0.5.2, this cannot be enforced, though there is a proposal to add
        // this feature: see https://github.com/ethereum/solidity/issues/4637
        uint256 _value; // default: 0
    }

    function current(Counter storage counter) internal view returns (uint256) {
        return counter._value;
    }

    function increment(Counter storage counter) internal {
        unchecked {
            counter._value += 1;
        }
    }

    function decrement(Counter storage counter) internal {
        uint256 value = counter._value;
        require(value > 0, "Counter: decrement overflow");
        unchecked {
            counter._value = value - 1;
        }
    }

    function reset(Counter storage counter) internal {
        counter._value = 0;
    }
}

// OpenZeppelin Contracts v4.4.1 (access/Ownable.sol)
/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

/// @notice Library to encode strings in Base64.
/// @author Solady (https://github.com/vectorized/solady/blob/main/src/utils/Base64.sol)
/// @author Modified from Solmate (https://github.com/transmissions11/solmate/blob/main/src/utils/Base64.sol)
/// @author Modified from (https://github.com/Brechtpd/base64/blob/main/base64.sol) by Brecht Devos - <brecht@loopring.org>.
library Base64 {
    /// @dev Encodes `data` using the base64 encoding described in RFC 4648.
    /// See: https://datatracker.ietf.org/doc/html/rfc4648
    /// @param fileSafe  Whether to replace '+' with '-' and '/' with '_'.
    /// @param noPadding Whether to strip away the padding.
    function encode(bytes memory data, bool fileSafe, bool noPadding)
        internal
        pure
        returns (string memory result)
    {
        /// @solidity memory-safe-assembly
        assembly {
            let dataLength := mload(data)

            if dataLength {
                // Multiply by 4/3 rounded up.
                // The `shl(2, ...)` is equivalent to multiplying by 4.
                let encodedLength := shl(2, div(add(dataLength, 2), 3))

                // Set `result` to point to the start of the free memory.
                result := mload(0x40)

                // Store the table into the scratch space.
                // Offsetted by -1 byte so that the `mload` will load the character.
                // We will rewrite the free memory pointer at `0x40` later with
                // the allocated size.
                // The magic constant 0x0230 will translate "-_" + "+/".
                mstore(0x1f, "ABCDEFGHIJKLMNOPQRSTUVWXYZabcdef")
                mstore(0x3f, sub("ghijklmnopqrstuvwxyz0123456789-_", mul(iszero(fileSafe), 0x0230)))

                // Skip the first slot, which stores the length.
                let ptr := add(result, 0x20)
                let end := add(ptr, encodedLength)

                // Run over the input, 3 bytes at a time.
                for {} 1 {} {
                    data := add(data, 3) // Advance 3 bytes.
                    let input := mload(data)

                    // Write 4 bytes. Optimized for fewer stack operations.
                    mstore8(0, mload(and(shr(18, input), 0x3F)))
                    mstore8(1, mload(and(shr(12, input), 0x3F)))
                    mstore8(2, mload(and(shr(6, input), 0x3F)))
                    mstore8(3, mload(and(input, 0x3F)))
                    mstore(ptr, mload(0x00))

                    ptr := add(ptr, 4) // Advance 4 bytes.
                    if iszero(lt(ptr, end)) { break }
                }
                mstore(0x40, add(end, 0x20)) // Allocate the memory.
                // Equivalent to `o = [0, 2, 1][dataLength % 3]`.
                let o := div(2, mod(dataLength, 3))
                // Offset `ptr` and pad with '='. We can simply write over the end.
                mstore(sub(ptr, o), shl(240, 0x3d3d))
                // Set `o` to zero if there is padding.
                o := mul(iszero(iszero(noPadding)), o)
                mstore(sub(ptr, o), 0) // Zeroize the slot after the string.
                mstore(result, sub(encodedLength, o)) // Store the length.
            }
        }
    }

    /// @dev Encodes `data` using the base64 encoding described in RFC 4648.
    /// Equivalent to `encode(data, false, false)`.
    function encode(bytes memory data) internal pure returns (string memory result) {
        result = encode(data, false, false);
    }

    /// @dev Encodes `data` using the base64 encoding described in RFC 4648.
    /// Equivalent to `encode(data, fileSafe, false)`.
    function encode(bytes memory data, bool fileSafe)
        internal
        pure
        returns (string memory result)
    {
        result = encode(data, fileSafe, false);
    }

    /// @dev Decodes base64 encoded `data`.
    ///
    /// Supports:
    /// - RFC 4648 (both standard and file-safe mode).
    /// - RFC 3501 (63: ',').
    ///
    /// Does not support:
    /// - Line breaks.
    ///
    /// Note: For performance reasons,
    /// this function will NOT revert on invalid `data` inputs.
    /// Outputs for invalid inputs will simply be undefined behaviour.
    /// It is the user's responsibility to ensure that the `data`
    /// is a valid base64 encoded string.
    function decode(string memory data) internal pure returns (bytes memory result) {
        /// @solidity memory-safe-assembly
        assembly {
            let dataLength := mload(data)

            if dataLength {
                let decodedLength := mul(shr(2, dataLength), 3)

                for {} 1 {} {
                    // If padded.
                    if iszero(and(dataLength, 3)) {
                        let t := xor(mload(add(data, dataLength)), 0x3d3d)
                        // forgefmt: disable-next-item
                        decodedLength := sub(
                            decodedLength,
                            add(iszero(byte(30, t)), iszero(byte(31, t)))
                        )
                        break
                    }
                    // If non-padded.
                    decodedLength := add(decodedLength, sub(and(dataLength, 3), 1))
                    break
                }
                result := mload(0x40)

                // Write the length of the bytes.
                mstore(result, decodedLength)

                // Skip the first slot, which stores the length.
                let ptr := add(result, 0x20)
                let end := add(ptr, decodedLength)

                // Load the table into the scratch space.
                // Constants are optimized for smaller bytecode with zero gas overhead.
                // `m` also doubles as the mask of the upper 6 bits.
                let m := 0xfc000000fc00686c7074787c8084888c9094989ca0a4a8acb0b4b8bcc0c4c8cc
                mstore(0x5b, m)
                mstore(0x3b, 0x04080c1014181c2024282c3034383c4044484c5054585c6064)
                mstore(0x1a, 0xf8fcf800fcd0d4d8dce0e4e8ecf0f4)

                for {} 1 {} {
                    // Read 4 bytes.
                    data := add(data, 4)
                    let input := mload(data)

                    // Write 3 bytes.
                    // forgefmt: disable-next-item
                    mstore(ptr, or(
                        and(m, mload(byte(28, input))),
                        shr(6, or(
                            and(m, mload(byte(29, input))),
                            shr(6, or(
                                and(m, mload(byte(30, input))),
                                shr(6, mload(byte(31, input)))
                            ))
                        ))
                    ))
                    ptr := add(ptr, 3)
                    if iszero(lt(ptr, end)) { break }
                }
                mstore(0x40, add(end, 0x20)) // Allocate the memory.
                mstore(end, 0) // Zeroize the slot after the bytes.
                mstore(0x60, 0) // Restore the zero slot.
            }
        }
    }
}

// from: https://eips.ethereum.org/EIPS/eip-2981#simple-summary
///
/// @dev Interface for the NFT Royalty Standard
///
interface IERC2981 is IERC165 {
    /// ERC165 bytes to add to interface array - set in parent contract
    /// implementing this standard
    ///
    /// bytes4(keccak256("royaltyInfo(uint256,uint256)")) == 0x2a55205a
    /// bytes4 private constant _INTERFACE_ID_ERC2981 = 0x2a55205a;
    /// _registerInterface(_INTERFACE_ID_ERC2981);

    /// @notice Called with the sale price to determine how much royalty
    //          is owed and to whom.
    /// @param _tokenId - the NFT asset queried for royalty information
    /// @param _salePrice - the sale price of the NFT asset specified by _tokenId
    /// @return receiver - address of who should be sent the royalty payment
    /// @return royaltyAmount - the royalty payment amount for _salePrice
    function royaltyInfo(
        uint256 _tokenId,
        uint256 _salePrice
    ) external view returns (
        address receiver,
        uint256 royaltyAmount
    );
}

// from: https://eips.ethereum.org/EIPS/eip-4906
// interface id for IERC4906: '0x49064906'
/// @title EIP-721 Metadata Update Extension
interface IERC4906 is IERC165, IERC721 {
    /// @dev This event emits when the metadata of a token is changed.
    /// So that the third-party platforms such as NFT market could
    /// timely update the images and related attributes of the NFT.
    event MetadataUpdate(uint256 _tokenId);

    /// @dev This event emits when the metadata of a range of tokens is changed.
    /// So that the third-party platforms such as NFT market could
    /// timely update the images and related attributes of the NFTs.
    event BatchMetadataUpdate(uint256 _fromTokenId, uint256 _toTokenId);
}

//===============================================
// ChekorX

interface efficaxInterface{
  function tokenURI(address, uint256) external view returns (string memory);
}

interface checksInterface{
  function svg(uint256) external view returns (string memory);
}

contract Chekorx is ERC721, IERC2981, IERC4906, Ownable {
  using Counters for Counters.Counter;

  struct Token {
    bytes18 salt;
    uint72 charity;
    uint16 flips;
    bool locked;
  }

  mapping(uint256 => Token) internal token;
  mapping(address => uint256) public mintsBy;

  Counters.Counter private supply;
  uint256 internal constant smileMod = 25;
  uint256 internal constant maxMints = 3;
  uint256 internal constant ownerMaxMints = 10;
  uint256 public mintFee = 0.01 ether;
  uint256 public mintCharity = 0.001 ether;
  uint256 public minFlipFee = 0.0002 ether;
  uint256 public constant maxSupply = 1000;
  bool public mintEnabled;
  bool public mintLocked;
  uint256 internal constant royaltyPercent = 5;
  string public externalURL = "https://nometa.online/chekorx";
  string public constant description =
    "An on-chain hack of FRESH HELL #1, ALGO BRO by XCOPY. Not to be used for making important life decisions. CC0";
  string internal constant nameStart = "ChekorX #";
  uint256 internal start;
  uint256 internal duration;
  bool public isLocked;
  /// @notice this address may send from charityBank via sendCharity()
  address public charityDelegate;
  /// @notice the current value of this contract earmarked for charity
  uint256 public charityBank;
  /// @notice the sum of all values withdrawn from this contract via sendCharity()
  uint256 public charityWithdrawn;
  address internal constant efficaxAddress = 0x323d8d9aB29731849345b97758eD4C1323A1e487;
  address internal constant checksAddress = 0x036721e5A769Cc48B3189EFbb9ccE4471E8A48B1;
  address internal constant freshHellAddress = 0xbB6EefD83DaaB02Fa1a697EB9D8495eEc9f9B384;

  constructor() payable ERC721('ChekorX','CoX') {
    charityDelegate = msg.sender;
  }

  efficaxInterface EfficaxContract = efficaxInterface(efficaxAddress);
  checksInterface ChecksContract = checksInterface(checksAddress);

  modifier tokenExists(uint256 _tokenId) {
    require(_exists(_tokenId), "token does not exist");
    _;
  }

  // interface id for IERC2981: '0x2a55205a'
  // interface id for IERC4906: '0x49064906'
  function supportsInterface(bytes4 interfaceId)
        public view override(IERC165, ERC721) returns (bool) {
    return
        interfaceId == type(IERC721).interfaceId ||
        interfaceId == type(IERC721Metadata).interfaceId ||
        interfaceId == type(IERC2981).interfaceId ||
        interfaceId == bytes4(0x49064906) ||
        super.supportsInterface(interfaceId);
  }

  receive() external payable {}

  /// @notice Locks the minting state. This cannot be undone.
  function lockMint() external onlyOwner {
    mintLocked = true;
  }

  /// @notice Toggles bool mintEnabled
  function toggleMint() external onlyOwner {
    require(!mintLocked, "minting has been locked");
    mintEnabled = !mintEnabled;
  }

  function safeIterate(uint256 _number) internal pure returns(uint256) {
    unchecked {
      return _number + 1;
    }
  }

  /// @notice mint given number of tokens
  /// @param _mints the number of tokens to mint. Limit 5 per address.
  function mint(uint256 _mints) external payable {
    require(mintEnabled, "minting has been disabled");
    address msgSender = msg.sender;
    uint256 mints = _mints;
    uint256 charity = mintCharity;
    unchecked {
      require((mintsBy[msgSender] + mints) <= maxMints, "mint limit of 3");
      require((supply.current() + mints) <= maxSupply, "exceeds max supply");
      require(msg.value == mints * mintFee, "invald value for tranasction");
      charityBank += mints * charity;
      mintsBy[msgSender] += mints;
    }
    for (uint256 count = 1; count <= mints; count = safeIterate(count)) {
      supply.increment();
      uint256 tokenId = supply.current();
      Token storage thisToken = token[tokenId];
      bytes32 salt = keccak256(abi.encodePacked(
                                          block.prevrandao,
                                          msgSender,
                                          tokenId,
                                          block.number,
                                          count
                                          ));
      thisToken.salt = bytes18(salt);
      thisToken.charity = uint72(charity);
      _safeMint(msgSender, tokenId);
    }
  }

  function ownerMint(uint256 _mints) external payable onlyOwner {
    // can mint when disabled
    address msgSender = owner();
    uint256 mints = _mints;
    uint256 charity = mintCharity;
    unchecked {
      require((mintsBy[msgSender] + mints) <= ownerMaxMints, "mint limit of 5");
      require((supply.current() + mints) <= maxSupply, "exceeds max supply");
      uint256 charityRequired = mints * charity;
      require(msg.value == charityRequired, "invald value for tranasction");
      charityBank += charityRequired;
      mintsBy[msgSender] += mints;
    }
    for (uint256 count = 1; count <= mints; count = safeIterate(count)) {
      supply.increment();
      uint256 tokenId = supply.current();
      Token storage thisToken = token[tokenId];
      bytes32 salt = keccak256(abi.encodePacked(
                                          block.prevrandao,
                                          msgSender,
                                          tokenId,
                                          block.number,
                                          count
                                          ));
      thisToken.salt = bytes18(salt);
      thisToken.charity = uint72(charity);
      _safeMint(msgSender, tokenId);
    }
  }

  /// @notice flip your token. Who knows what you'll get
  /// @param _tokenId the id of the token to be flipped
  function flip(uint256 _tokenId) external payable tokenExists(_tokenId) {
    address msgSender = msg.sender;
    uint256 tokenId = _tokenId;
    uint256 msgValue = msg.value;
    // testing minFlipFee cost - i don't think there's a difference
    require(msgValue >= minFlipFee, "incorrect value sent");
    require(msgSender == ownerOf(tokenId), "not token owner");
    require(!flipped(), "flip is broken. please stand by");
    Token storage thisToken = token[tokenId];
    require(!thisToken.locked, "token has been locked");
    bytes32 salt = keccak256(abi.encodePacked(
                                        block.prevrandao,
                                        msgSender,
                                        block.number,
                                        tokenId
                                        ));
    thisToken.salt = bytes18(salt);
    unchecked {
      thisToken.flips += 1;
      thisToken.charity += uint72(msgValue);
      charityBank += msgValue;
    }

    emit MetadataUpdate(tokenId);
  }

  /// @notice permanently lock a token from being flipped again.
  /// @param _tokenId the id of the token to be locked
  function lockFlip(uint256 _tokenId) external tokenExists(_tokenId) {
    uint256 tokenId = _tokenId;
    require(msg.sender == ownerOf(tokenId), "not token owner");
    token[tokenId].locked = true;
  }

  /// @notice called with the sale price to determine how much royalty is owed and to whom.
  /// @param _tokenId the NFT asset queried for royalty information
  /// @param _salePrice the sale price of the NFT asset specified by _tokenId
  /// @return receiver address of who should be sent the royalty payment
  /// @return royaltyAmount the royalty payment amount for _salePrice
  function royaltyInfo(uint256 _tokenId, uint256 _salePrice)
                external view tokenExists(_tokenId)
                returns(address receiver, uint256 royaltyAmount) {
    return (owner(), _salePrice * royaltyPercent / 100);
  }

  function setIt(uint256 _start, uint256 _duration) external onlyOwner {
    require(!isLocked, "not fooling anyone");
    require(_duration <= 432000, "over max duration");
    start = _start;
    duration = _duration;
  }

  function lockIt() external onlyOwner {
    start = 0;
    isLocked = true;
  }

  function flipped() internal view returns(bool) {
    if (start == 0) {return false;}
    if (isLocked) {return false;}
    uint256 time = block.timestamp;
    uint256 end;
    unchecked{end = start + duration;}
    return (time > start && time < end);
  }

  function emitUpdate() external onlyOwner {
    emit BatchMetadataUpdate(1, supply.current());
  }

//======================================================================
// artwork

  /// @notice this is external only for calldata slicing.
  function decodeAlgoBro(bytes calldata _bstring)
                  external pure returns(string memory) {
    string memory svg64 = string(_bstring[107:57067]);
    return string(Base64.decode(svg64));
  }

  /// @notice get the metadata string of ALGO BRO from 0xdiid's Efficax contract
  function getAlgoBro() internal view returns(string memory) {
    return this.decodeAlgoBro(abi.encodePacked(
                                EfficaxContract.tokenURI(freshHellAddress, 1)
                                ));
  }

  /// @notice get the svg string for checks #1
  function getChecks() internal view returns(string memory) {
    return ChecksContract.svg(1);
  }

  function generateColours(bytes calldata _salt)
              pure external returns(string memory, string memory) {
    return (string.concat(
                'rgb(',
                Strings.toString(uint8(_salt[31])), ',',
                Strings.toString(uint8(_salt[30])), ',',
                Strings.toString(uint8(_salt[29])), ')'
                ),
            string.concat(
                'rgb(',
                Strings.toString(uint8(_salt[28])), ',',
                Strings.toString(uint8(_salt[27])), ',',
                Strings.toString(uint8(_salt[26])), ')'
                ));
  }

  /// @notice this will determine your fate.
  function getResult(uint256 _tokenId) internal view
              returns(uint256, string memory, string memory) {
      uint256 salt = uint144(token[_tokenId].salt);
      uint256 result;
      if (salt % smileMod == 0) {
        result = 2;
      } else {
        result = salt % 2;
      }
      (string memory colour1, string memory colour2) = this.generateColours(abi.encodePacked(salt));
      return (result, colour1, colour2);
    }

  /// @notice builds and returns the svg as a string
  /// @param _tokenId If you've come this far, you know what this means.
  function generateSVG(uint256 _tokenId)
                public view tokenExists(_tokenId) returns(string memory) {
    (uint256 result, string memory colour1, string memory colour2) = getResult(_tokenId);
    string memory class;
    string memory groupBlock;
    string memory groups;
    string memory extraStyle;
    string memory checks;
    if (flipped()) {
      result = 2;
    }
    if (result == 1) {
      class = 'mouth';
      groupBlock = string.concat(
              '<use class="tick1" href="#Tick1"/>',
              '<use class="tick2" href="#Tick2"/>',
              '<use class="tick3" href="#Tick3"/>',
              '<use class="tick4" href="#Tick4"/>',
            '</g>'
            );
      groups = string.concat(
            '<g transform="scale(6) translate(-1250 -1450)">',
            groupBlock,
            '<g transform="scale(3) translate(-2330 -1200)">',
            groupBlock
            );
      extraStyle = string.concat(
            '.mouth{fill:',colour1,'}',
            '.tick1{fill:',colour1,';animation-delay:0.5s}',
            '.tick2{fill:#4fc7f9}',
            '.tick3{fill:#51ff82}',
            '.tick4{fill:',colour2,'}'
            );
    } else {
      class = 'teeth';
      groupBlock = string.concat(
              '<use class="frame1" href="#X1"/>',
              '<use class="frame2" href="#X2"/>',
              '<use class="frame3" href="#X3"/>',
              '<use class="frame4" href="#X4"/>',
            '</g>'
            );
      groups = string.concat(
            '<g transform="scale(5) translate(-1868 -920)">',
            groupBlock,
            '<g transform="scale(1) translate(-1068 -120)">',
            groupBlock
            );

      extraStyle = string.concat(
            '.teeth{fill:',colour1,'}',
            '.frame1{fill:',colour1,';animation-delay:0.5s}',
            '.frame2{fill:#ff42d5}',
            '.frame3{fill:#ff0c0c}',
            '.frame4{fill:',colour2,'}'
            );
    }
    if (result == 2) {
      groups = string.concat(
            '<g transform="scale(1) translate(-800 -1000)">',
              '<use href="#MOUTH"/>',
              '<use class="frame1" transform="scale(1 0.8) translate(0 560)" href="#MOUTH"/>',
              '<use href="#TEETH"/>',
              '<use class="tick3" transform="translate(5 10)" href="#TEETH"/>',
            '</g>',
            '<g transform="scale(1.5) rotate(20) translate(-1000 -1120)">',
            groupBlock,
            '<g transform="scale(2) rotate(-15) translate(-1768 -820)">',
            groupBlock,
            '<use class="tick1" transform="scale(3) translate(220 130)" href="#check"/>',
            '<use class="tick2" transform="scale(3) translate(223 132)" href="#check"/>',
            '<use class="tick3" transform="scale(2) rotate(20) translate(778 -15)" href="#check"/>',
            '<use class="tick4" transform="scale(2) rotate(20) translate(780 -13)" href="#check"/>'
            );
      checks = getChecks();
    }
    string memory style = string.concat(
            '<style type="text/css">',
            '.border { fill:', colour2, '; animation: borderAnimation 0.16s infinite alternate;}',
            extraStyle,
            '@keyframes borderAnimation {0%,50%{fill:', colour1, ';}51%,100%{fill:', colour2, ';}}',
            '</style>'
            );
    return string.concat(
            '<svg style="background-color:#111111" ',
                'viewBox="0 0 800 800" xmlns="http://www.w3.org/2000/svg">',
              '<svg class="inner" width="100%" height="100%" viewBox="0 0 2000 2000">',
                '<defs>',
                  getAlgoBro(),
                  checks,
                '</defs>',
                '<rect width="100%" height="100%" fill="#0c1b2c"/>',
                '<rect x="4%" y="4%" class="border" width="92%" height="92%"/>',
                '<rect x="8%" y="8%" class="',class, '" width="84%" height="84%"/>',
                '<rect x="10%" y="10%" class="head" width="80%" height="80%"/>',
                groups,
                style,
              '</svg>',
            '</svg>'
            );
  }

  /// @notice returns the token svg image as a data uri
  /// @param _tokenId the id of the token
  function imageURI(uint256 _tokenId) public view tokenExists(_tokenId) returns(string memory) {
    return string.concat(
              "data:image/svg+xml;base64,",
              Base64.encode(bytes(generateSVG(_tokenId)))
              );
  }

  function genMetaData(uint256 _tokenId) internal view returns(string memory) {
    uint256 tokenId = _tokenId;
    return string.concat(
            'data:application/json;base64,',
            Base64.encode(abi.encodePacked(
                        '{"name":"',
                        nameStart,
                        Strings.toString(tokenId),
                        '","description":"',
                        description,
                        '",',
                        '"external_url":"',
                        externalURL,
                        '","attributes":',
                        genAttribs(tokenId),
                        ',"image":"',
                        imageURI(tokenId),
                        '"}'
                        )));
  }


  function genAttribs(uint256 _tokenId) internal view returns(string memory) {
    uint256 tokenId = _tokenId;
    Token memory thisToken = token[tokenId];
    (uint256 result, string memory colour1, string memory colour2) = getResult(tokenId);
    string memory _flip;
    if (result == 0) {
      _flip = "x";
    } else if (result == 1) {
      _flip = "check";
    } else {
      _flip = "smile";
    }
    string memory lock;
    if (thisToken.locked) {
      lock = "locked";
      } else {
        lock = "unlocked";
      }
    string memory started;
    if (flipped()) {
      started = '"},{"trait_type":"IPittyTheFoo","value":"you';
      lock = "locked";
    }
    string memory flips = Strings.toString(thisToken.flips);
    string memory charity;
    unchecked {
      charity = string.concat(
                    '},{"trait_type":"charityGwei)","value":',
                    Strings.toString(thisToken.charity / 1e9)
                    );
    }
    return string.concat(
                    '[{"trait_type":"flip","value":"',
                    _flip,
                    '"},{"trait_type":"colour1","value":"',
                    colour1,
                    '"},{"trait_type":"colour2","value":"',
                    colour2,
                    '"},{"trait_type":"flips","value":',
                    flips,
                    charity,
                    '},{"trait_type":"state","value":"',
                    lock,
                    started,
                    '"}]'
                    );
  }

  // @notice returns the number of tokens in the collection
  function totalSupply() external view returns (uint256) {
      return supply.current();
  }

  /// @notice returns metadata for the token
  /// @param _tokenId the id of the token
  function tokenURI(uint256 _tokenId)
        public view virtual override tokenExists(_tokenId) returns (string memory) {
  return genMetaData(_tokenId);
  }

  /// @notice Set the external address for metadata. onlyOwner
  /// @param _url The new external url.
  function setExternalURL(string calldata _url) external onlyOwner {
      externalURL = _url;
  }

  /// @notice update the minimum fee required for flipping a token
  function setFlipFee(uint256 _fee) external onlyOwner {
    minFlipFee = _fee;
  }

  /// @notice set an address as the charity delegate
  // only the charity delegate can send the funds from charityBank
  function setCharityDelegate(address _delegate) external onlyOwner {
    charityDelegate = _delegate;
  }

  /// @notice withdraw contract balance minus charityBank
  function withdraw() external onlyOwner {
    uint256 withdrawValue = address(this).balance - charityBank;
    (bool success, ) = payable(msg.sender).call{value: withdrawValue}("");
    require(success);
  }

  /// @notice send funds from the charityBank to an address
  /// @param _charity the address to send funds to
  /// @param _amount the value to send. 0 will send the full value of charityBank
  function sendCharity(address _charity, uint256 _amount) external {
    require(msg.sender == charityDelegate, "sender not delegated");
    uint256 bank = charityBank;
    uint256 amount = _amount;
    require(amount <= bank, "not enough funds in bank");
    if (amount == 0) {
      amount = bank;
    }
    unchecked {
      charityWithdrawn += amount;
      charityBank -= amount;
    }
   (bool success, ) = payable(_charity).call{value: amount}("");
   require(success);
  }
 }